

Paper 094-30

SAS 8 Is Seen in the Rearview Mirror: Generate Statistical Graphics Using ODS in SAS®9

Wei Cheng, ISIS Pharmaceuticals, Inc., Carlsbad, CA


ABSTRACT

In SAS V8, the SAS Output Delivery System (ODS) in SAS/STAT greatly enhances the ability to manage statistical tables. In SAS 9.1, ODS Statistical Graphics (Experimental) enables SAS users to generate statistical graphics as easily as tables. This paper describes how to create statistical graphics automatically from a number of SAS/STAT procedures, as well as how to customize the layout and the appearance of statistical graphics using ODS templates and ODS styles.

INTRODUCTION

The SAS system continues to improve the graphics generated by SAS/STAT procedures. SAS/STAT procedures could only generate line printer plots before SAS Version 6.12. The plotting facility in a number of SAS/STAT procedures had been upgraded to high resolution plots since SAS Version 7. The high resolution graphics can be saved in graphics catalogs and controlled by the SAS/GRAPH GOPTIONS statement. The Product-Limit survival function estimate curves in Figure 1 were generated by the PROC LIFETEST procedure. Useful statistical information can be put into titles and footnotes by macro variables. You don't have any control over the layout of the graphics area, such as legend for treatment group, y-axis label and tick marks. For information about how to generate the high resolution plots from PROC LIFETEST, refer to this author's WUSS 10 paper (Cheng, 2002). If you really want to generate customized curves, you have to save analysis results in an output data set, display them with a SAS/GRAPH procedure and use the Annotate facility.

FIGURE 1



In SAS 9, a number of SAS/STAT procedures can use the ODS to create statistical graphics automatically. The new facility is referred to as ODS Statistical Graphics (or ODS Graphics for short). The layout and appearance of ODS Graphics are controlled by ODS templates and styles. You can use PROC TEMPLATE to create your statistical graphics templates and generate statistical graphics from DATA step, or you can use PROC TEMPLATE to customize the statistical graphics generated by SAS/STAT procedures. Unlike the high resolution plots in SAS Version 8, in SAS 9 you can control every part on the graphics area.

In this paper, PROC LIFETEST will be used as an example to demonstrate how to generate graphics using ODS Graphics, and how to customize graphics using the TEMPLATE procedure. The programs can be adapted to other SAS/STAT procedures which support ODS Graphics with the same principle.

ODS Statistical Graphics are experimental in SAS 9.1, the appearance of graphs currently produced by procedures, as well as the syntax (procedure options and graph template language) are subject to change in a future release.

GENERATING STATISTICAL GRAPHICS

The ODS GRAPHICS statement is specified to request ODS Graphics in addition to the tabular output:


EXAMPLE 1

```
ods graphics on / reset imagename = "sugigraph" ;
ods html path = "C:\sugi30" gpath = "C:\sugi30\graph" (url = "graph/")
      file = "sugi.html" style = statistical;

proc lifetest data = exposed;
  time days * status(0);
  strata treatment;
run;

ods html close;
ods graphics off;
```

The data set EXPOSED is taken from the "Getting Started" section of Chapter 40, "The LIFETEST Procedure", in the SAS OnlineDoc 9.1.3. Figure 2 displays the graph of the product-limit survival function estimate versus survival time, as well as the summary statistics, generated from the preceding program:



If you want to view the other graphical displays of the survival distribution and related function estimates that PROC LIFETEST can produce, you can specify the PLOTS = option in the SURVIVAL statement:


EXAMPLE 2

```
ods graphics on / reset imagename = "sugigraph" ;
ods html path = "C:\sugi30" gpath = "C:\sugi30\graph" (url = "graph/")
      file = "sugi.html" style = statistical;

proc lifetest data = exposed;
  time days * status(0);
  strata treatment;
  survival plots = ( survival, density, epb, hazard, loglogs, logsurv,
                    hwb, cl, stratum);
run;

ods html close;
ods graphics off;
```

Figure 3 to 9 are other statistical graphics generated by the preceding program, and include a panel of plots for each stratum, a plot of the survival functions estimates for all strata, a plot of the Hall-Wellner bands for all strata, a plot of the equal precision bands for all strata, a plot of negative log of the estimated survival functions, and a plot of the log of the negative log of the estimated survival functions:



CUSTOMIZING LAYOUT OF ODS GRAPHICS

Statistical graphics are controlled by graph templates, which are written in the graph template language. The default graph templates supplied by SAS should suffice in most common cases of SAS procedures. However, ODS Graphics enables you to modify the default template to customize the graphics when modifications are needed.

The default graph templates `Stat.Lifetest.Graphics.ProductLimit` and `Stat.Lifetest.Graphics.ProductLimitPanel` will be used to demonstrate how to customize the layout of statistical graphics with templates.

SWAPPING THE LOCATIONS OF FREQUENCY AND PERCENTAGE

In the summary statistics displayed with the Product-Limit survival function estimates curve in Figure 2, the frequency of Event and Censored are displayed inside parentheses. If you want to swap the locations of frequency and percentage, and display the percentage inside parentheses, you can modify the portion which specifies the summary statistics in the template:

EXAMPLE 3

```
ods path work.mytmpl(update) sashelp.tmplmst(read);
proc template;
  define statgraph Stat.Lifetest.Graphics.ProductLimit;
```


```

[SNIP] layout gridded / rows=1 halign=left;
        entry NEVENT1 / padleft=0 padright=0;
        entry " (" / padleft=0 padright=0;
        entry PCTEVENT1 / format=2.0 halign=right padright=0;
        entry "%)" / padleft=0;
        endlayout;
        layout gridded / rows=1 halign=left;
        entry NCENSORED1 / padleft=0 padright=0 format = 3.0;
        entry " (" / padleft=0 padright=0;
        entry PCTCENSORED1 / format=2.0 halign=right padright=0;
        entry "%)" / padleft=0;
        endlayout;
[SNIP] entry NOBS2;
        layout gridded / rows=1 halign=left;
        entry NEVENT2 / padleft=0 padright=0;
        entry " (" / padleft=0 padright=0;
        entry PCTEVENT2 / format=2.0 halign=right padright=0;
        entry "%)" / padleft=0;
        endlayout;
        layout gridded / rows=1 halign=left;
        entry NCENSORED2 / padleft=0 padright=0 format = 3.0;
        entry " (" / padleft=0 padright=0;
        entry PCTCENSORED2 / format=2.0 halign=right padright=0;
        entry "%)" / padleft=0;
        endlayout; [SNIP]

end;
run;

```

The resulting Product-Limit survival function estimates, after submitting the preceding program and Example 1, is shown in Figure 10:



SWAPPING THE LOCATIONS OF LEGEND FOR P-VALUE AND TREATMENT GROUP

This example is a continuation of Example 3. The objective is to swap the locations of the legend which specify p-value and treatment group. In the "Product-Limit Survival" template definition, the options HALIGN and VALIGN specify the horizontal and vertical alignment of the legend:

EXAMPLE 4

```

proc template;
  define statgraph Stat.Lifetest.Graphics.Productlimit;
    [SNIP]
    DiscreteLegend "Survival" / halign=left valign=bottom across=1
                                border=true;
    layout gridded / rows=1 halign=right valign=top border=true;


```

```

[SNIP]
end;
run;

```

To display the modifications, shown in Figure 11, submit the modified template definition and then resubmit the program in Example 1.



CHANGING TICK MARKS AND LABEL OF Y-AXIS

Sometimes you don't like the default label "Survival Probability" of the y-axis and the default tick marks on the y-axis which are 0, 0.2, 0.4, 0.6, 0.8, and 1. This example, which is a continuation of Example 4, illustrates how to change the tick marks and the label of y-axis:


EXAMPLE 5

```

proc template;
  define statgraph Stat.Lifetest.Graphics.Productlimit;
    [SNIP]
    layout overlay / yaxisopts=( label="My Label for Y Axis"
      ticks=( 0 .1 .2 .3 .4 .5 .6 .7 .8 .9 1.0 ) );
    [SNIP]
  end;
run;

```

The resulting Product-Limit survival curve, after submitting the preceding statements and the statements in Example 1, is shown in Figure 12:



CHANGING THE SYMBOL VALUE FOR THE CENSORED OBSERVATIONS


This example, which is a continuation of Examples 3 - 5, illustrates how to change the symbol value for the censored observations.

You can use the **MARKERSYMBOL=** option to specify the symbol value to mark the censored observations. By default, the symbol is a "+" sign. You can change the symbol to an asterisk (*):

EXAMPLE 6

```
proc template;
  define statgraph Stat.Lifetest.Graphics.Productlimit;
    [SNIP]
    scatter y=CENSORED x=TIME / group=STRATUM
          index=STRATUMNUM
          markersymbol=star
          markersize=StatGraphData:markersize;
    [SNIP]
  end;
run;
```

The result of the change, after submitting the modified template definition and the program in Example 1, is displayed in Figure 13:

**ADDING GRID LINES TO Y-AXIS**


The next example illustrates how to add grid lines to the y-axis. It is a continuation of Example 6.

You can add the grid lines by specifying **YGRID = TRUE**, and the degree of transparency by specifying **GRIDTRANSPARENCY = 0.5** as half transparent:

EXAMPLE 7

```
proc template;
  define statgraph Stat.Lifetest.Graphics.Productlimit;
    [SNIP]
    layout overlay / yaxisopts=( label="My Label for Y Axis"
          ticks=( 0 .1 .2 .3 .4 .5 .6 .7 .8 .9 1.0 ) )
          ygrid = 1 gridtransparency = 0.5;
    stepplot y=SURVIVAL x=TIME / group=STRATUM
          index=STRATUMNUM
          name="Survival";
    scatter y=CENSORED x=TIME / group=STRATUM
          index=STRATUMNUM
          markersymbol=star
          markersize=StatGraphData:markersize;
    [SNIP]
  end;
run;
```

The result of these modifications, after submitting the modified template definition and the program in Example 1, is displayed in Figure 14:



SHOWING CURVES ONLY

The last example of modifications to the Stat.Lifetest.Graphics.ProductLimit template illustrates how to get rid of the summary statistics (the box at the bottom) altogether from the graph of Kaplan-Meier estimates. This is desirable if the graph is going to appear in a report where the summary statistics has already been tabulated.


If you only need to have the Kaplan-Meier curves on your graph, then only the statements to generate the curves need to be included in the template:

EXAMPLE 8

```
proc template;
define statgraph Stat.Lifetest.Graphics.Productlimit;
dynamic Logrank;
  layout gridded;
    layout lattice / rows=1 columns=1;
      sidebar / align=top;
        layout overlay / padbottom=5;
          entrytitle "Product-Limit Survival Function Estimates";
          endlayout;
        endsidebar;
      layout overlay / yaxisopts=( label="Survival Probability"
        ticks=( 0 .2 .4 .6 .8 1.0 ) );
        stepplot y=SURVIVAL x=TIME / group=STRATUM index=STRATUMNUM
          name="Survival";
        scatter y=CENSORED x=TIME / group=STRATUM index=STRATUMNUM
          markersymbol=plus markersize=
            StatGraphData:markersize;
        DiscreteLegend "Survival" / hAlign=right vAlign=top across=1
          border=true;
      layout gridded / rows=1 halign=left valign=bottom border=true;
        if (LOGRANK<.0001)
          entry "Logrank p" / padright=0;
        else
          entry "Logrank p=" / padright=0;
        endif;
        entry LOGRANK / padleft=0 format=pvalue6.4;
      endlayout;
    endlayout;
  endlayout;
end;
run;
```

After submitting the modified template definition and the program in Example 1, only the Product-Limit Survival Function Estimates curves are displayed in Figure 15:

FIGURE 15



MODIFYING PANELS

The next example illustrates how to modify the panel of plots for each stratum in Figure 3 and Figure 4. The original panel consists of three plots and a summary statistics text box. The template that defines the original panel is "Stat.Lifetest.Graphics.ProductLimitPanel". In Example 9, the panel is modified so that it includes only the Product-Limit survival curve and the summary statistics with 25 and 75 percentile statistics removed. The ROWS=1 and COLUMNS=2 options in the LAYOUT LATTICE statement specify a 1 x 2 lattice. The LAYOUT statements for components 3 and 4, percentiles 25 and 75 are deleted.

EXAMPLE 9

```
proc template;
  define statgraph Stat.Lifetest.Graphics.ProductLimitPanel;
    [SNIP]
    layout lattice / rows=1
                  columns=2
                  vrange=unionall
                  hrange=unionall;
    [SNIP]
    entry "50";
    layout gridded / rows=1
                  halign=left;
    [SNIP]
  end;
run;
```

After submitting the modified template definition and the program in Example 2, you can get a simplified panel, which is shown in Figure 16 and Figure 17:

FIGURE 16





FIGURE 17



CUSTOMIZING APPEARANCE OF ODS GRAPHICS

If you have modified a style to customize the appearance of ODS tables, you can modify a style to customize the appearance of ODS Graphics in the same way. Example 10 shows how to enlarge the fonts for the axis value (tick mark) labels and axis labels if you want to emphasize them in the graph.

EXAMPLE 10

```
ods path work.mytmpl(update) sashelp.tmplmst(read);

proc template;
  define style mystyle;
    parent = styles.statistical;
    replace GraphFonts /
      'GraphDataFont' = ("Arial",8pt)
      'GraphFootnoteFont' = ("Arial",8pt,Bold)
      'GraphTitleFont' = ("Arial",8pt,Bold)
      'GraphValueFont' = ("Arial",16pt,Bold)
      'GraphLabelFont' = ("Arial",18pt,Bold);
  end;
run;

ods graphics on / reset imagename = "sugigraph";


ods html path = "C:\sugi30"
         gpath = "C:\sugi30\graph" (url = "graph/")
         file = "sugi.html" style = mystyle;

proc lifetest data = exposed;
  time days * status(0);
  strata treatment;
run;

ods html close;

ods graphics off;
```

Submit the preceding program to display the Product-Limit survival curve created with your modified style, as shown in Figure 18:

**CONCLUSION**

ODS Statistical Graphics is an exciting new facility provided by SAS 9. It integrates the graphic display with the tabular output. It automates the production of statistical graphics which become more and more important in modern statistical analysis, while maintaining the ability to customize the displays by modifying the ODS templates and ODS styles. SAS users are encouraged to try the experimental release of ODS Graphics and provide feedback to the SAS Institute. It is not recommended for production work at the release of SAS 9.1.

REFERENCES

1. SAS Institute, Inc. (2002-2004), "Statistical Graphics Using ODS – Experimental" *SAS/STAT User's Guide*, SAS OnlineDoc® 9.1.3, Chapter 15, Cary, NC: SAS Institute, Inc.
2. SAS Institute, Inc. (2002-2004), "The LIFETEST Procedure" *SAS/STAT User's Guide*, SAS OnlineDoc® 9.1.3, Chapter 40, Cary, NC: SAS Institute, Inc.
3. SAS Institute, Inc. (2005), "TEMPLATE Procedure: Creating ODS Statistical Graphics Output (Experimental)" < <http://support.sas.com/rnd/base/topics/statgraph/proctemplate/>> (Jan 21, 2005).
4. SAS Institute, Inc. (2005), "ODS Statistical Graphics (Experimental) and ODS Styles: Usage and Reference" < <http://support.sas.com/rnd/base/topics/statgraph/v91StatGraphStyles.htm>> (Jan 21, 2005).
5. Cheng, Wei (2002), "Perform survival analysis for clinical trials using ODS" *Proceedings of the Tenth Western Users of SAS Software Conference*.

ACKNOWLEDGMENTS

I would like to thank Bob Rodriguez and Ying So of SAS Institute for their review and comments.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Wei Cheng,

Isis Pharmaceuticals, Inc.,

2292 Faraday Ave., Carlsbad, CA 92008

Email: wcheng@isisph.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.